Maple-assisted proof of formula for A196141

Robert Israel

21 Jan 2019

There are $5^6 = 15625$ configurations for a 2×3 sub-array, but not all can arise.

We encode these configurations as lists in the order $\begin{bmatrix} x_1 & x_2 & x_3 \\ x_4 & x_5 & x_6 \end{bmatrix}$ Let $b_0 = 3$, $b_1 = 0$, $b_2 = 4$, $b_3 = 1$, $b_4 = 2$. If $x_i = j$, then the number of neighbours of site i in the sub-array that have value b_j must be either j or $j - 1$.

```
> b[0]:= 3: b[1]:= 0: b[2]:= 4: b[3]:= 1: b[4]:= 2:
  goodconfig:= proc(x) local t;
     t:= numboccur(b[x[1]], [x[2],x[4]]);
     if t < x[1]-1 or t > x[1] then return false fi;
     t:= numboccur(b[x[2]], [x[1],x[3],x[5]]);
     if t < x[2]-1 or t > x[2] then return false fi;
     t:= numboccur(b[x[3]], [x[2],x[6]]);
     if t < x[3]-1 or t > x[3] then return false fi;
     t:= numboccur(b[x[4]], [x[1],x[5]]);
     if t < x[4]-1 or t > x[4] then return false fi;
     t:= numboccur(b[x[5]], [x[2],x[4],x[6]]);
     if t < x[5]-1 or t > x[5] then return false fi;
     t:= numboccur(b[x[6]], [x[3],x[5]]);
     if t < x[6]-1 or t > x[6] then return false fi;
     true
  end proc:
  Configs:= select(goodconfig, [seq(convert(5^6+i,base,5)[1..6],i=
  0..5^6-1)]):
  nops(Configs);
```

$$76 \quad (1)$$

There are 76 allowed configurations.

Consider the 76×76 transition matrix T with entries $T_{ij} = 1$ if the first two rows of a 3×3 sub-array could be in configuration i while the last two rows are in configuration j, and 0 otherwise. The following code computes it.

```
> Compatible:= proc(i,j) local Xi,Xj,k;
   Xi:= Configs[i]; Xj:= Configs[j];
   if Xi[4..6] <> Xj[1..3] then return 0 fi;
   if numboccur(b[Xi[4]],[Xi[1],Xi[5],Xj[4]]) <> Xi[4] then return
  0 fi;
   if numboccur(b[Xi[5]],[Xi[2],Xi[4],Xi[6],Xj[5]]) <> Xi[5] then
  return 0 fi;
   if numboccur(b[Xi[6]],[Xi[3],Xi[5],Xj[6]]) = Xi[6] then 1 else 0
  fi;
  end proc:
  T:= Matrix(76,76,Compatible):
```

Thus for $n \geq 2$ $a(n) = \dfrac{u^T T^{n-2} v}{2}$ where u is a column vector with 1 for configurations whose first row could be a top row, 0 otherwise, and similarly v has 1 for configurations whose second row could be a bottom row.

```
> u:= Vector(76,proc(i) local x; x:= Configs[i];
     if numboccur(b[x[1]], [x[2],x[4]])=x[1]
     and numboccur(b[x[2]],[x[1],x[3],x[5]]) = x[2]
     and numboccur(b[x[3]],[x[2],x[6]]) = x[3] then 1 else 0 fi
  end proc):
  v:= Vector(76,proc(i) local x; x:= Configs[i];
     if numboccur(b[x[4]], [x[1],x[5]]) = x[4]
     and numboccur(b[x[5]],[x[2],x[4],x[6]]) = x[5]
     and numboccur(b[x[6]],[x[3],x[5]]) = x[6] then 1 else 0 fi
   end proc) :
```

To check, here are the first few entries of our sequence (apart from a_1, which doesn't really fit the pattern, although it does work with the recurrence).

```
> Tv[0]:= v:
  for n from 1 to 32 do Tv[n]:= T . Tv[n-1] od:
> A:= [seq(u^%T . Tv[n],n=0..32)];
```

$A := [8, 7, 26, 49, 85, 178, 348, 683, 1349, 2688, 5319, 10498, 20818, 41206, 81574, 161646, 320215, 634294, 1256481, 2489029, 4930656, 9767642, 19350237, 38333645, 75940498, 150441579, 298031468, 590414638, 1169642000, 2317123308, 4590345948, 9093722915, 18015156930]$ **(2)**

Now here is the empirical recurrence formula. It says that $u^T T^n Q(T)\, v = 0$ for all nonnegative integers n, where Q is the following polynomial.

```
> n:= 'n': empirical:= a(n) = 3*a(n-1) -2*a(n-2) +a(n-3) -3*a(n-4)
  +3*a(n-5) -a(n-6) -7*a(n-8) +a(n-9) +4*a(n-10) +2*a(n-11) +6*a
  (n-12):
  Q:= unapply(add(coeff((lhs-rhs)(empirical),a(n-i))*t^(12-i),i=0.
  .12),t);
```

$$Q := t \mapsto t^{12} - 3\,t^{11} + 2\,t^{10} - t^9 + 3\,t^8 - 3\,t^7 + t^6 + 7\,t^4 - t^3 - 4\,t^2 - 2\,t - 6 \quad \textbf{(3)}$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto t^{21} - 4\,t^{20} + 6\,t^{19} - 5\,t^{18} + t^{17} + 3\,t^{16} - 2\,t^{15} - t^{14} + t^{13} + 3\,t^{12} + 2\,t^{11} + 4\,t^{10} - 21\,t^9 + 16\,t^8 + 12\,t^7 - 12\,t^6 - 16\,t^4 - 12\,t^3 \quad \textbf{(4)}$$

$Q(t)$ should be a factor of $P(t)$.

```
> R:= unapply(normal(P(t)/Q(t)),t);
```

$$R := t \mapsto \left(t^6 - t^5 + t^4 + t^3 - 2\,t^2 + 2\,t + 2\right) t^3 \quad \textbf{(5)}$$

The complementary factor $R(t)$ has degree 9.

Now we want to show that $c(n) = u^T\, Q(T)\, T^n\, v = 0$ for all $n \geq 0$. This will certainly satisfy the recurrence

$$\sum_{i=3}^{9} r_i\, c(i+n) = \sum_{i=3}^{9} r_i u^T\, Q(T)\, T^{n+i} v = u^T\, Q(T)\, R(T)\, T^n\, v = u^T\, P(T)\, T^n\, v = 0$$

where r_i are the coefficients of $R(t)$. To show all $c(n) = 0$ it suffices to show $c(0) = \ldots = c(8) = 0$.

```
> uT[0]:= u^%T:
  for n from 1 to 12 do uT[n]:= uT[n-1] . T od:
  uQ:= add(coeff(Q(t),t,n)*uT[n],n=0..12):
```

```
seq(uQ . Tv[i],i = 0 .. 8);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0 \quad (6)$$

This completes the proof.